MAS

Display



(HP-41CX, Hewlett Packard 1983 and DM41X, SwissMicros 2020)

Overview[1]

Programs MASD and MASA are mastermind code-breaking games. The user must find a code which consists of either 4 or 5 digits, representing colours from the original game invented in 1970 by Mordecai Meirowitz, an Israeli postmaster and telecommunications expert. In these programs the user is first prompted to find a 4-digit or a 5-digit code. In the first case, digits are randomly selected from a range 1-6 and in the second case from 1-8. Duplicate digits are not allowed.

When the user makes the first guess, e.g. 5216, the calculator will determine the match and display the following information:

```
5216 3.2 1
```

in which

```
5216 = the code entered by the user
3.2  = 3 digits are OK, 2 of these are on the right position
1    = the number of guesses the user has made
```

Program MASD uses decimal calculation to determine the match of an user-given guess of the secret code. This version is a bit faster than the alpha-digit based version MASA.

Program MASA checks digits in the alpha register to determine the match.

Both programs have a similar way of creating and storing a secret code. The method makes use of the registers R01-R06 or R01-R08 for a 4-digits respectively 5-digits code. If a digit is randomly chosen, its position in the secret code is stored in the register. If the digit is not present in the secret code, the corresponding register value remains 0. The example code 5216 is stored in the registers as follows: R01: 3, R02: 2, R03: 0, R04: 0, R05: 1, R06: 4

After entering 4 or 5 for the type of game to run, the startup of the programs shows the initial display consisting of zeroes:

```
0000 0.0 0
```

which is to have consistency and ease of use. The user can now enter the first guess.

[1] *This program is copyright and is supplied without representation or warranty of any kind. The author assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof*

Example (1): MASD

KEYSTROKES	DISPLAY	COMMENTS
[XEQ] [ALPHA] MASD [ALPHA]	GAME 4/57	Start; choose 4- or 5-digits code
4 [R/S]	0000 0,0 0	Initialisation done, make first guess
1234 [R/S]	1234 3,0 1	3 Digits ok, none on right position
2345 [R/S]	2345 3,0 2	Second guess, same result
3451 [R/S]	3451 3,2 3	Third guess, 3 ok, 2 right position
3652 [R/S]	3652 3,3 4	Fourth guess, 3 ok and right position
3152 [R/S]	3152 5x	Secret code revealed in 5 turns
[R/S]	GAME 4/57	Try another game

Example (2): MASA

KEYSTROKES	DISPLAY	COMMENTS
[XEQ] [ALPHA] MASA [ALPHA]	GAME 4/57	Start; choose 4- or 5-digits code
4 [R/S]	0000 0.0 0	Initialisation done, make first guess
1234 [R/S]	1234 2.0 1	2 Digits ok; 5 and 6 must be present
5612 [R/S]	5612 3.1 2	Second guess, 3 ok, 1 right position
5361 [R/S]	5361 3.1 3	Third guess, 3 ok, 1 right position
5126 [R/S]	5126 3.0 4	Fourth guess, 3 ok
4562 [R/S]	4562 3.3 5	Fifth guess, 3 ok and right position
3562 [R/S]	3562 6x	Secret code revealed in 6 turns
[R/S]	GAME 4/57	Try with a 5-digit code
5 [R/S]	00000 0.0 0	Ready, make first guess
12345 [R/S]	12345 4.0 1	First guess gives 4 matches
23784 [R/S]	23784 3.0 2	Second guess, 3 ok, 1 right position
34271 [R/S]	34271 4.1 3	Third guess, 3 ok, 1 right position
41257 [R/S]	41257 4.1 4	Fourth guess, 3 ok
61478 [R/S]	61478 2.0 5	Fifth guess, 3 ok and right position
31527 [R/S]	31527 5.3 6	Sixth guess, 3 ok and right position
35127 [R/S]	35127 7x	Secret code revealed in 6 turns
[R/S]	GAME 4/57	Choose for another game

Program Listing

The listing of the programs is given below with 2 XROM functions SEED and RNDM on lines 3 and 13 resp. 3 and 15. These can be taken as explicit calls to other programs in memory or to XROM functions, for example the CCD module. If replaced by these XROM functions (as in the RAW and TXT file) the total number of byte savings is 8. So, program MASD would be 148 instead of 156 bytes and program MASA would be 152 instead of 160 bytes. If not available, programs must be in place for these functions.

The listing of program MASD ("decimal" based) is given below:

```
01▪LBL "MASD"
02 CLRG
03 XEQ "SEED"
04 CF 29
05 "GAME 4/5?"
06 PROMPT
07 CLA
08 STO 00
09 STO 09
10 LBL 02
11 >"0"
12 LBL 03
13 XEQ "RNDM"
14 RCL 00
15 ST+ X
16 DSE X
17 *
18 INT
19 ,
20 X#NN?
21 GTO 03
22 RCL 09
23 STO IND Z
24 DSE 09
25 GTO 02
26 GTO 01
27 LBL 04
28 ,
29 STO 09
30 RDN
31 CLA
32 FIX 0
33 ARCL X
34 RCL 00
35 ISG 10
36 LBL 05
37 X<>Y
38 10
39 ST/ Y
40 X<>Y
41 INT
42 LASTX
43 FRC
44 ST* Z
45 RDN
46 RCL IND Y
47 X=0?
48 GTO 06
49 R^
50 1
51 ST+ 09
52 RDN
53 X#Y?
54 GTO 08
55 ,1
56 ST+ 09
57 RDN
58 LBL 08
59 RDN
60 LBL 06
61 X<> T
62 DSE X
63 GTO 05
64 RCL 00
65 10
66 /
67 RCL 09
68 FRC
69 LBL 01
70 >" "
71 X=Y?
72 GTO 07
73 FIX 1
74 ARCL 09
75 FIX 0
76 >" "
77 ARCL 10
78 PROMPT
79 GTO 04
80 LBL 07
81 ARCL 10
82 >"X"
83 FIX 5
84 SF 29
85 AVIEW
86 END

(156 bytes)
```

and for MASA ("alpha" based) shown here:

```
01▪LBL "MASA"
02 CLRG
03 XEQ "SEED"
04 FIX 00
05 CF 29
06 "GAME 4/5?"
07 PROMPT
08 STO 13
09 2
10 ST* Y
11 -
12 STO 12
13 RCL 13
14 LBL 03
15 XEQ "RNDM"
16 RCL 12
17 *
18 INT
19 1
20 +
21 X<>Y
22 X<NN?
23 GTO 03
24 STO IND Y
25 DSE X
26 GTO 03
27 DSE 00
28 LBL 04
29 .
30 STO 10
31 STO 11
32 RDN
33 RCL 13
34 10^X
35 X<>Y
36 +
37 CLA
38 ARCL X
39 ATOX
40 LASTX
41 RCL 12
42 ISG 00
43 LBL 01
44 STO Y
45 48
46 +
47 POSA
48 1
49 +
50 X=0?
51 SF 29
52 RCL IND Y
53 FS?C 29
54 GTO 02
55 X=0?
56 GTO 02
57 ISG 10
58 X#Y?
59 X=Y?
60 ISG 11
61 LBL 02
62 X<> Z
63 DSE X
64 GTO 01
65 >" "
66 RCL 13
67 RCL 11
68 X=Y?
69 GTO 05
70 ARCL 10
71 >"."
72 ARCL 11
73 >" "
74 ARCL 00
75 PROMPT
76 GTO 04
77 LBL 05
78 ARCL 00
79 >"X"
80 FIX 05
81 SF 29
82 AVIEW
83 END

(160 bytes)
```

Registers, Labels and Flags

REGISTERS	COMMENTS
R00	Number of digits (MASD)
R01-R08	Secret code (positions)
R09	Matches & Positions (MASD)
R10	Number of guesses (MASD)
R00	Number of guesses (MASA)
R10	Matches (MASA)
R11	Positions (MASA)
R12	Max. digit value (MASA)
R13	Number of digits (MASA)

LABELS MASD	COMMENTS
LBL01	Finish start display
LBL02	Initialse start display
LBL03	Randomise secret code
LBL04	Show input guess
LBL05	Check input guess
LBL06	Display guess outcome
LBL07	Display end result
LBL08	Display guess outcome
LABELS MASA	
LBL01	Check input guess
LBL02	Display guess outcome
LBL03	Randomise secret code
LBL04	Show input guess
LBL05	Display end result

FLAGS	COMMENTS
29	Disable decimal separator

Downloads

The RAW/TXT format of the program is available via the website: MAS (in zip file).